


# Topic 7

## Introduction to Verilog HDL

# Descriptions of a Half Adder




- Addition of two single bits A, B
- Based on the operations it performs, a truth table can be built

A	B	Sum	Carry
0	0	0	0
0	1	1	0
1	0	1	0
1	1	0	1


- Derive Boolean functions (sum-of-minterms) from the truth table for both outputs

$$\begin{aligned}\text{Sum} &= A'B + AB' = m_1 + m_2 = \Sigma (1, 2) \\ &= (A+B)(A'+B') = M_0 \cdot M_3 = \Pi (0, 3)\end{aligned}$$

$$\begin{aligned}\text{Carry} &= AB = m_3 \\ &= (A+B)(A+B')(A'+B) \\ &= M_0 \cdot M_1 \cdot M_2 \\ &= \Pi (0, 1, 2)\end{aligned}$$



# Alternative Description of a Half Adder Hardware Description Language (HDL)



module name      module ports

```
module Add_half (sum, c_out, a, b);  
  input      a, b;  
  output     sum, c_out;      ← declaration of port modes
```

Keywords

```
  wire       c_out_bar;      ← declaration of internal signal
```

这三个函数  
平行执行

```
    xor        (sum, a, b);  
    nand       (c_out_bar, a, b);  
    not        (c_out, c_out_bar);  
  endmodule
```

instantiation of pre-defined primitive gates

Same variable indicates connection


# Full Adder Implemented with Half Adder

```
module Add_full (sum, c_out, a, b, c_in); // parent module
  input    a, b, c_in;
  output   c_out, sum;
  wire     w1, w2, w3;

  Add_half M1 (w1, w2, a, b);           // child module
  Add_half M2 (sum, w3, w1, c_in);       // child module
  or (c_out, w2, w3);                   // primitive instantiation
endmodule
```

M1 M2  
也是  
平行  
执行

Module instance name



# 4-bit Carry-Ripple Adder

**module** Add\_rca\_4 (sum, c\_out, a, b, c\_in); 行波进位加法器

**output** [3: 0] sum; 一个四位的bus作为sum


**output** c\_out;

**input** [3: 0] a, b;

**input** c\_in;

定义bus的方法

**wire** c\_in2, c\_in3, c\_in4;



Add\_full M1 (sum[0], c\_in2, a[0], b[0], c\_in);

Add\_full M2 (sum[1], c\_in3, a[1], b[1], c\_in2);

Add\_full M3 (sum[2], c\_in4, a[2], b[2], c\_in3);

Add\_full M4 (sum[3], c\_out, a[3], b[3], c\_in4);

**endmodule**

# 16-bit Carry-Ripple Adder (Top)

```
module Add_rca_16 (sum, c_out, a, b, c_in);
```


```
  output [15:0] sum;
```

```
  output c_out;
```

```
  input [15:0] a, b;
```

```
  input c_in;
```

16 bit buses



```
wire c_in4, c_in8, c_in12, c_out;
```

```
Add_rca_4 M1 (sum[3:0], c_in4, a[3:0], b[3:0], c_in);
```


```
Add_rca_4 M2 (sum[7:4], c_in8, a[7:4], b[7:4], c_in4);
```

```
Add_rca_4 M3 (sum[11:8], c_in12, a[11:8], b[11:8], c_in8);
```

```
Add_rca_4 M4 (sum[15:12], c_out, a[15:12], b[15:12], c_in12);
```

```
endmodule
```

## 2-bit Comparator (unsigned numbers)



A1	A0	B1	B0	<	>	=
0	0	0	0	0	0	1
0	0	0	1	1	0	0
	...				...	
1	1	0	0	0	1	0
1	1	0	1	0	1	0
1	1	1	0	0	1	0
1	1	1	1	0	0	1


- Boolean equations:

$$A\_lt\_B = A1' B1 + A1' A0' B0 + A0' B1 B0$$

$$A\_gt\_B = A1 B1' + A0 B1' B0' + A1 A0 B0'$$

$$A\_eq\_B = A1' A0' B1' B0' + A1' A0 B1' B0 + A1 A0 B1 B0 + A1 A0' B1 B0'$$

# Gate-level Schematic






# Comparator – Structural Model

```
module compare_2_str (A_lt_B, A_gt_B, A_eq_B, A0, A1, B0, B1);  
  input      A0, A1, B0, B1;  
  output     A_lt_B, A_gt_B, A_eq_B;
```

```
  wire       w1, w2, w3, w4, w5, w6, w7;
```

```
  or    (A_lt_B, w1, w2, w3);  
  nor   (A_gt_B, A_lt_B, A_eq_B);  
  and   (A_eq_B, w4, w5);  
  and   (w1, w6, B1);  
  and   (w2, w6, w7, B0);  
  and   (w3, w7, B1, B0);  
  not   (w6, A1);  
  not   (w7, A0);  
  xnor  (w4, A1, B1);  
  xnor  (w5, A0, B0);  
endmodule
```

May be implicitly declared



# Comparator – RTL Model

```
module compare_2 (A_lt_B, A_gt_B, A_eq_B, A1, A0, B1, B0);  
  input      A1, A0, B1, B0;  
  output     A_lt_B, A_gt_B, A_eq_B;  
  
  assign A_lt_B = (~A1)&B1 | (~A1)&(~A0)&B0 | (~A0)&B1&B0;  
  assign A_gt_B = A1&(~B1) | A0&(~B1)&(~B0) | A1&A0&(~B0);  
  assign A_eq_B = (~A1)&(~A0)&(~B1)&(~B0) | (~A1)&A0&(~B1)&B0  
                | A1&A0&B1&B0 | A1&(~A0)&B1&(~B0);  
  
endmodule
```

!! assign !! important keyword

- Continuous assignment statements
- All concurrently executed



# Comparator – Alternative RTL Model

```
module compare_2_logic (A_lt_B, A_gt_B, A_eq_B,  
                        A1, A0, B1, B0);  
  
  input      A1, A0, B1, B0;  
  output     A_lt_B, A_gt_B, A_eq_B;  
  
  assign     A_lt_B = ({A1, A0} < {B1, B0});  
  assign     A_gt_B = ({A1, A0} > {B1, B0});  
  assign     A_eq_B = ({A1, A0} == {B1, B0});  
endmodule
```



Concatenation of A1 and A0

# Comparator – Behavioral Model

```
module compare_2_algo (A_lt_B, A_gt_B, A_eq_B, A,B);  
  input      [1:0] A, B;  2-bit bus  
  output     A_lt_B, A_gt_B, A_eq_B;  
  reg        A_lt_B, A_gt_B, A_eq_B;  
  always @ (A or B)  Cyclic statement triggered upon @condition  
  begin  
    A_lt_B = 0;  
    A_gt_B = 0;  
    A_eq_B = 0;  
    if (A==B) A_eq_B = 1;  
    else if (A>B) A_gt_B = 1;  
    else A_lt_B = 1;  
  end  
endmodule
```

不是真的register,只是一种type

only when A or B is changed, this always statement is triggered 在某些情况下触发

Destination variables inside always must be register (reg)

# RTL Alternative of 16-bit Adder


```
module add_rtl_16 (sum, c_out, a, b, c_in);  
  input [15:0] a, b;  
  input c_in;  
  output [15:0] sum;  
  output c_out;
```

don't know the detail of the circuit

```
  assign {c_out, sum} = a + b + c_in;  
endmodule
```

变成了一个17bit的

Concatenation of c\_out and sum



# 2-to-1 MUX

```
module MUX_2_1 (Out, I0, I1, Sel);
```

```
  input I0, I1, Sel;
```

```
  output Out; when any of the three is changed, it will be triggered
```

```
  reg Out;
```

```
  always @(I0, I1, Sel) begin 也叫sensitively select 当只有一个执行语块的时候begin end不必须
```

```
    case (Sel)
```

```
      1'b0: Out = I0;
```


```
      1'b1: Out = I1;
```

```
      default Out = 0;
```

```
    endcase
```

```
  end
```

```
endmodule
```



*1' means one bit binary number b0 and b1*

*Like **switch...case...** in C/C++*

*总共有0,1,x,z四种output, 所以需要default*

*错误分析: **always** @ (Sel)  
case(Sel)*

*1'b0: Out = I0;*

*1'b1: Out = I1;*

*因为当I0变了以后, 整个case不会被触发, 所以out的值不变*

# Flip-Flop – Modeling Clock Behavior

```
module D_ff (q, data_in, clk);  
  input data_in, clk;  
  output q;
```

```
  reg q;
```


```
  always @ (posedge clk)
```

```
  begin
```

```
    q <= data_in;
```

```
  end
```

```
endmodule
```



Rising edge of


Non-blocking assignment statement

这是一个 assigning statement

# Synchronous Control Input

```
module D_ff (q, data_in, clk, syn_rst);  
  input data_in, clk, syn_rst;  
  output q;  
  
  reg q;  
  
  always @ (posedge clk)  
  begin  
    if (syn_rst == 1) q <= 0;  
    else q <= data_in;  
  end  
endmodule
```


Synchronous reset





# Flip-Flops with Control Inputs


- D flip flop with active low synchronous Clear



# Asynchronous Control Input


```
module D_ff (q, data_in, clk, asyn_rst);  
  input data_in, clk, asyn_rst;  
  output q;  
  
  reg q;  
  
  always @ (posedge clk or posedge asyn_rst)  
  begin  
    if (asyn_rst == 1) q <= 0;  
    else q <= data_in;  
  end  
endmodule
```

Asynchronous active  
high reset



# Asynchronous Control Input

- D flip flop with active low asynchronous Clear



# Registers

- GENERAL RULE: A variable will be synthesized as a flip-flop when its value is assigned synchronously with an edge of a signal

```
module D_reg4 (Data_in, clock, reset, Data_out);  
  input  [3:0] Data_in;  
  input      clock, reset;  
  output [3:0] Data_out;  
  reg  [3:0] Data_out;  
  always @ (posedge reset or posedge clock)  
    if (reset == 1'b1) Data_out <= 4'b0;  
    else  
      Data_out <= Data_in;  
endmodule
```

from 0 to 1


4 bits of 0

同理, 32'b0 32bits of 1  
!然而! 4'b1 是0001

want an active high

既然已经知道时posedge  
为什么还要check 1

目的是为了  
告诉执行的  
这一个不是clock




# Hardware Description Language (HDL)

- An HDL is a language that describes the hardware of digital systems in a textual form
  - Can describe digital system specified at different levels of abstraction
- There are many HDLs,
  - two most popular IEEE standards: VHDL and Verilog HDL;
  - other IEEE standards: SystemC, SystemVerilog, HandleC...

# Different HDLs


- VERILOG

```
module and_gate (y, x1, x2);  
    input    x1, x2;  
    output   y;  
  
    and (y , x1, x2);  
endmodule
```



- VHDL

```
entity and_gate is  
    port (x1, x2: in bit;  
          y: out bit);  
end and_gate;  
architecture data_flow of  
    and_gate is  
begin  
    y <= x1 and x2;  
end data_flow;
```




# Hardware Description Language (HDL)

- What is HDL used for?
  - Another way to describe digital circuit
  - Quick functional verification – simulation
  - Quick virtual to real circuit conversion – synthesis
- Advantages in digital design
  - Most ‘reliable’ design process, with minimum cost and time-to-market (TTM)
  - Reduce fault penalty!

# HDL Modeling

- The HDL model specifies a relationship (scheduling rule) between input signals and output signals





# Summary: Verilog Module Structure

**module** the\_design ( ... );

declarations: ports, constants, variables, events

declarations: tasks and functions

} Declaration

instantiations of predefined modules

continuous assignment: assign y = ...

behavioral statements (**initial**, **always**) {

procedural (blocking) assignment

procedural nonblocking assignment

procedural-continuous assign

event trigger

task calls

function calls

} Implementation

}

**endmodule**

# Parameterized Module

```
module Param_Examp (y_out, a, b);  
    parameter size = 8, delay = 15;  
    output [size-1:0] y_out;  
    input    [size-1:0] a, b;  
    wire     [size-1:0] #delay y_out; // net transport delay  
    // Other declarations, instantiations,  
    // and behavior statements go here.  
endmodule
```

有点类似于宏定义

- Verilog allows parameters to be overridden on an instance basis and by hierarchical dereferencing.

# Parameter Annotation

```
module modXnor (y_out, a, b);  
  parameter size = 8, delay = 15;  
  output [size-1:0] y_out;  
  input [size-1:0] a, b;  
  wire [size-1:0] #delay y_out = a~^b; //bitwise xnor  
endmodule
```

```
module Param;  
  modXnor G1 (y1_out, b1, c1); //Instantiation with  
                                //default parameters  
  modXnor #(4,5) G2 (y2_out, b2, c2); //Uses size = 4,  
                                       //delay = 5  
endmodule
```

表示要改变，并重新定义一个modXnor

这句表示改变parameter size as 4 and delay as 5. 好处在于不用改变开头的定义

# Procedural Assignments

- Blocking procedural assignment (=)
- Non-blocking Procedural Assignment (<=)
- Left Hand Side must be reg data type


# Nonblocking Procedural Assignment

- Evaluation and schedule of the RHS of an assignment is not blocked by the activity of preceding statements in a sequential activity flow
  - All nonblocking procedural assignments evaluate their RHS at the same time
  - Evaluated values are scheduled to assigned to LHS concurrently
- Assignment operator:  $\leq$
- Syntax:  $\langle lvalue \rangle \leq [\text{timing control}] \langle expression \rangle;$
- The outcome of executing a sequential list of nonblocking assignments is independent of the order of the list.

# Blocking Vs. Nonblocking Assignment


- The listed order affects the outcome of blocking assignments

如果是blocking,那么  
必须是逐条执行  
执行完第一条后  
D已经变成E了  
这一块意味着  
E经过一轮变换来到A



D <= E;      A <= B;  
C <= D;      B <= C;  
B <= C;      C <= D;  
A <= B;      D <= E;

or



# Unwanted Latch


- Incomplete case statement or conditional branch results in latches, even for combinational circuit

- Example:

```
always @( sel_a or sel_b or data_a or data_b)
  case ({sel_a, sel_b})
    2'b10: y_out = data_a;
    2'b01: y_out = data_b;
  endcase
```

问题在于这个没有default. 这个例子中缺少了00 11以及x与z的关系

- Synthesis result:




# Unwanted Latch – Fixed

- Fix

```
always @( sel_a or sel_b or data_a or data_b)  
  case ({sel_a, sel_b})  
    2'b10: y_out = data_a;  
    2'b01: y_out = data_b;  
    default y_out = 0;  
  endcase
```


- Synthesis result:





# Unwanted Latch

```
module incomplete_and (y, a1, a2);  
  input a1, a2;  
  output y;  
  reg y;  
  always @(a1 or a2)  
    if ({a2, a1} == 2'b11)      y = 1; else  
    if ({a2, a1} == 2'b01)      y = 0; else  
    if ({a2, a1} == 2'b10)      y = 0;  
endmodule
```




# Unwanted Latch – Fixed

```
module incomplete_and (y, a1, a2);  
  input a1, a2;  
  output y;  
  reg y;  
  always @(a1 or a2) begin  
    y = 0;  
    if ({a2, a1} == 2'b11)      y = 1; else  
    if ({a2, a1} == 2'b01)      y = 0; else  
    if ({a2, a1} == 2'b10)      y = 0;  
  end  
endmodule
```

当什么都没发生的时候，y就是0

instead of completing if and else. we give the initial value



# Reference

- Advanced Digital Design with Verilog HDL, 2/e, Michael Ciletti, 2010, ISBN: 978-0136019282
- IEEE Standard for Verilog HDL, [www.ieee.org](http://www.ieee.org)